RR-52

ISSN 0252-1075
Research Report No. RR-052

Contributions from

Indian Institute of Tropical Meteorology

A FORTRAN Algorithm For Divergent and Rotational Wind Fields

By

M. K. TANDON

PUNE - 411 008
INDIA

NOVEMBER 1992

A FORTRAN Algorithm For Divergent and Rotational Wind Fields

M.K.Tandon

Indian Institute of Tropical Meteorology, Pune 411 008, INDIA. RR-52

ABSTRACT

A FORTRAN algorithm for computing divergent and rotational wind fields from the combined total field is presented in this study. Reconstruction of the total wind field from the computed divergent and rotational wind fields is found to be exact at all points. This technique enables us to express the grid point data in terms of exact analytical functions whose differentials can be exactly evaluated. This feature of this technique is instrumental in the exact reproduction of original data from the computed components. Code of this ANSI FORTRAN-77 algorithm is listed in text as APPENDIX-B. The FORTRAN software for this algorithm was developed and tested on ND-560/Cx supermini computer system installed at IITM.

Key Words : FORTRAN , Potential , Divergent , Rotational, Correlation , Troposphere , Supermini , Wind, Fields, Covariance.

1. Introduction.

Following the analytical technique discussed in Tandon (1991), it is proposed to present an algorithm for expressing any 2-D field in terms of its divergent and rotational components in the following form.

$$\vec{V} = \vec{V}_{\psi} + \vec{V}_{\chi} \tag{1}$$

where $\vec{V}_{\psi}$ is rotational (non divergent) wind field and $\vec{V}_{\chi}$ is the divergent (irrotational) wind field. Eq.(1) in conjunction with the implications of nondivergence and irrotationality of a field yields the following Helmholtz equation.

$$\vec{V} = \hat{k} \times \nabla \psi + \nabla \chi \qquad (2)$$

where ∇ (del) is the 2-dimensional horizontal gradient operator, ψ is the stream function, $\hat{k}$ is the unit vector along the normal to the horizontal and χ is the velocity potential. This problem has been dealt with in some of the earlier studies including Stephens and Johnson (1978) , Bijlsma et. al. (1986), Sardeshmukh and Hoskins (1987). Bijlsma et. al. included a nondivergent and irrotational wind component in the Helmholtz equation, which was subsequently grouped with irrotational wind component to render partitioning unique. They solved equations $\xi = \nabla^2 \psi$ and $D = \nabla^2 \chi$ simultaneously using the relaxation technique . Stephens and Johnson (1978) has discussed this problem using a different procedure. Their method was based on discrete Fourier transform of consistent difference approximations of the potential equations wherein extra care is to be taken for reconstructing the boundary values of original field. Sardeshmukh and Hoskins (1987) derived the divergent flow from the rotational flow which in turn implies some sort of inter dependence between divergent and rotational components of any 2-D distribution (field). We, however, wish to tackle this problem using the technique of truncated double fourier series as explained in Tandon (1991). In this algorithm the regeneration of the original field from the computed divergent & rotational components is found to be exact at all points including the boundary values without making any extra provision for the boundary points. In this analysis we have studied (i) how the qualitative variations in one flow influence the qualitative variations in the other and (ii) the comparative contributions of these divergent and rotational flows in the total field. In this technique, we are able to express the grid point data in terms of exact functions (unlike other referenced techniques) whose derivatives can be evaluated exactly. This feature of this algorithm is responsible for the exact regeneration of the original grid point data from the computed components.

2. Methodology

Following the methodology of Tandon (1991) any function F : F(x,y) can be expressed in the following form.

$$F(x,y) = \sum_{n=0}^{N} \sum_{m=0}^{M} \left[\left\{ F_{nm}^{cc} \operatorname{Cos}(mkx) + F_{nm}^{cs} \operatorname{Sin}(mkx) \right\} \operatorname{Cos}(nly) + \left\{ F_{nm}^{sc} \operatorname{Cos}(mkx) + F_{nm}^{ss} \operatorname{Sin}(mkx) \right\} \operatorname{Sin}(nly) \right] \quad (3)$$

The first (second) sub and super script refers to meridional (zonal) directions respectively. Meridional and zonal directions are denumerated by "y" and "x" respectively. Terms F_{nm}^{cc}, F_{nm}^{cs}, F_{nm}^{sc} and F_{nm}^{ss} of Eq.(3) are coefficients of the terms Cos(nly)Cos(mkx) , Cos(nly)Sin(mkx) , Sin(nly)Cos(mkx) and Sin(nly)Sin(mkx) with F_{nm}^{cs} not equal to F_{nm}^{sc}. N and M are the meridional and zonal wave number truncations. (0,L) and (0,D) are zonal and meridional periods with L and D as the corresponding wavelengths ($k=2\pi/L$, $l = 2\pi/D$). Expansion coefficients F_{nm}^{cc}, F_{nm}^{cs}, F_{nm}^{sc} and F_{nm}^{ss} are computed as in Tandon (1991).

We first compute divergence and vorticity from the zonal and meridional components (U,V) of total wind "W" using the following relational expressions.

$$D = \frac{\partial U}{\partial x} + \frac{\partial V}{\partial y} \quad (4.1)$$

$$\zeta = \frac{\partial V}{\partial x} - \frac{\partial U}{\partial y} \tag{4.2}$$

where U and V are as defined in Eq.(3) with F=(U,V). Comparison of these expressions of D and   with their corresponding forms obtained from Eq.(3) gives the expansion coefficients of D and ζ in terms of the expansion coefficients of U and V. Later D and ζ are recomputed from streamfunction (ψ') and wind potential (χ) using the following relations.

$$D = \nabla^2 \chi \tag{5.1}$$

$$\zeta = \nabla^2 \psi \tag{5.2}$$

where ψ and χ are as expressed in Eq.(3) with F=(U,V).

Comparision of these expressions of D and ζ with the corresponding obtained from Eq.(3) gives us the expansion coefficients of D and ζ in terms of the expansion coefficients of ψ and χ . On equating the expansion coefficients of divergence and vorticity viz. D^{cc}_{nm} , D^{cs}_{nm} , D^{sc}_{nm} , D^{ss}_{nm} , ζ^{cc}_{nm} , ζ^{cs}_{nm} , ζ^{ss}_{nm} and ζ^{sc}_{nm} ,obtained by these two different procedures, we get the expansion coefficients of ψ and χ in terms of the expansion coefficients of U and V in the following form.

$$\chi^{cc}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (mk\, U^{cs}_{nm} + nl\, V^{sc}_{nm}) \tag{6.1}$$

$$\chi^{cs}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (-mk\, U^{cc}_{nm} + nl\, V^{ss}_{nm}) \quad (6.2)$$

$$\chi^{sc}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (mk\, U^{ss}_{nm} - nl\, V^{cc}_{nm}) \quad (6.3)$$

$$\chi^{ss}_{nm} = (m^2k^2 + n^2l^2)^{-1} (mk\, U^{sc}_{nm} + nl\, V^{ss}_{nm}) \quad (6.4)$$

$$\psi^{cc}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (mk\, V^{cs}_{nm} + nl\, U^{sc}_{nm}) \quad (7.1)$$

$$\psi^{cs}_{nm} = (m^2k^2 + n^2l^2)^{-1} (mk\, V^{cc}_{nm} + nl\, U^{ss}_{nm}) \quad (7.2)$$

$$\psi^{sc}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (mk\, V^{ss}_{nm} + nl\, U^{cc}_{nm}) \quad (7.3)$$

$$\psi^{ss}_{nm} = -(m^2k^2 + n^2l^2)^{-1} (-mk\, V^{sc}_{nm} + nl\, U^{cs}_{nm}) \quad (7.4)$$

Computation of zonal ($U_\psi = -\frac{\partial\psi}{\partial y}$, $U_\chi = \frac{\partial\chi}{\partial x}$) and meridional ($V_\psi = \frac{\partial\psi}{\partial x}$, $V_\chi = \frac{\partial\chi}{\partial y}$) components of rotational and divergent wind fields from ψ and χ fields and their subsequent forms obtained from Eq.(3) by replacing F by (U_ψ ,V_ψ ,U_χ ,V_χ) gives the expansion coefficients of U_ψ ,U_χ ,V_ψ ,V_χ as

$$\left(U^{cc}_{\psi nm}, U^{cs}_{\psi nm}, U^{sc}_{\psi nm}, U^{ss}_{\psi nm}\right) = \left(-nl\psi^{sc}_{nm}, -nl\psi^{ss}_{nm}, nl\psi^{cc}_{nm}, nl\psi^{cs}_{nm}\right) \quad (8.1)$$

$$\left(V^{cc}_{\psi nm}, V^{cs}_{\psi nm}, V^{sc}_{\psi nm}, V^{ss}_{\psi nm}\right) = \left(mk\psi^{cs}_{nm}, -mk\psi^{cc}_{nm}, mk\psi^{ss}_{nm}, -mk\psi^{sc}_{nm}\right) \quad (8.2)$$

$$\left(U^{cc}_{\chi nm}, U^{cs}_{\chi nm}, U^{sc}_{\chi nm}, U^{ss}_{\chi nm}\right) = \left(mk\chi^{cs}_{nm}, -mk\chi^{cc}_{nm}, mk\chi^{ss}_{nm}, -mk\chi^{sc}_{nm}\right) \quad (9.1)$$

$$\left(V^{cc}_{\chi nm}, V^{cs}_{\chi nm}, V^{sc}_{\chi nm}, V^{ss}_{\chi nm}\right) = \left(nl\chi^{sc}_{nm}, nl\chi^{ss}_{nm}, -nl\chi^{cc}_{nm}, -nl\chi^{cs}_{nm}\right) \quad (9.2)$$

Using these relations defining the expansion coefficients of zonal and meridional components of rotational and divergent wind fields in terms of the expansion coefficients of ψ and χ . Grid point distributions of U_ψ ,U_χ ,V_ψ ,V_χ are computed as in Tandon (1991).

3. Recovery

It is found that rotational and divergent wind fields , U_ψ ,V_ψ ,U_χ and V_χ, obtained by using the set of Eqs.(3),(6), (7),(8) and (9) regenarate the original total fields exactly. In particular we shall see the reconstruction of total U from U_ψ and U_χ i.e. how U_ψ and U_χ add up to give original total U. Using the set of Eqs.(6),(7),(8) and (9) we get the expansion coefficients of the quantity $(U_\psi + U_\chi)$ as under.

$$(U_\psi + U_\chi)^{cc}_{nm} = mk\chi^{cs}_{nm} - nl\psi^{sc}_{nm} \tag{10.1}$$

$$(U_\psi + U_\chi)^{cs}_{nm} = -mk\chi^{cc}_{nm} - nl\psi^{ss}_{nm} \tag{10.2}$$

$$(U_\psi + U_\chi)^{sc}_{nm} = mk\chi^{ss}_{nm} + nl\psi^{cc}_{nm} \tag{10.3}$$

$$(U_\psi + U_\chi)^{ss}_{nm} = -mk\chi^{sc}_{nm} + nl\psi^{cs}_{nm} \tag{10.4}$$

Eqs.(10.1),(10.2),(10.3) and (10.4) in conjunction with the set of Eqs.(6) and (7) yields the following.

$$\left[(U_\psi + U_\chi)^{cc}_{nm}, (U_\psi, U_\chi)^{cs}_{nm}, (U_\psi, U_\chi)^{sc}_{nm}, (U_\psi, U_\chi)^{ss}_{nm}\right]$$

$$= (U^{cc}_{nm}, U^{cs}_{nm}, U^{sc}_{nm}, U^{ss}_{nm}) \tag{11}$$

$$\text{i.e. } U_\psi + U_\chi = U$$

Eq.(11) shows that the expansion coefficients of the quantity $(U_\psi + U_\chi)$ are identical to those of zonal component ,U,of total field implying thereby that U_ψ and U_χ reconstruct total U exactly. Similarly, it can be easily seen that V_ψ and V_χ also reconstruct total V exactly. On using the set of Eqs.(6),(7), (8) and (9) we get the expansion coefficients of the quantities

$$\left[\frac{\partial U_\psi}{\partial x} + \frac{\partial V_\psi}{\partial y}\right] \text{ (say "A") and } \left[\frac{\partial V_\chi}{\partial x} - \frac{\partial U_\chi}{\partial y}\right] \text{ (say "B") as}$$

$$(A_{nm}^{cc}, A_{nm}^{cs}, A_{nm}^{sc}, A_{nm}^{ss}) = (0,0,0,0) \tag{12.1}$$

$$(B_{nm}^{cc}, B_{nm}^{cs}, B_{nm}^{sc}, B_{nm}^{ss}) = (0,0,0,0) \tag{12.2}$$

Eqs.(12.1) and (12.2) show that the expansion coefficients of the quantities "A" and "B" are all zeros implying thereby the following results.

$$\frac{\partial U_\psi}{\partial x} + \frac{\partial V_\psi}{\partial y} = 0 \tag{13.1}$$

$$\frac{\partial V_\chi}{\partial x} - \frac{\partial U_\chi}{\partial y} = 0 \tag{13.2}$$

i.e. (U_ψ, V_ψ) and (U_χ, V_χ) computed using Eqs.(3),(6),(7),

(8) and (9) constitute nondivergent (rotational) and irrotational (divergent) components of the total field expressed by Eq.(1).

4. Example

For analysing the results of the aforedescribed algorithm, we choose the monsoon seasonal (June-August) normal zonal and meridional winds at 200 mb level from the published charts of Newell et. al. (1972) as the zonal and meridional components of the total wind as represented by Eq.(1). Winds at 200 mb are subjectively interpolated on a 5 degree square grid in the domain (30 S,30 N),(60 E,145 E) giving 13 and 18 as the number of grid points along the meridional (latitudinal "y") and the zonal (longitudinal "x") directions respectively.Expansion coefficients are computed using "Efficient Fourier Transform" (EFT) method. Wave number truncations used in summations are $N = 7$ and $M = 10$. Isopleths of observed U, U_ψ, U_χ, observed V, V_ψ, V_χ (all in units of meter/sec) for the scanned domain are shown in Figs.1,2,3,4,5 and 6, respectively. Isoline maps of $U_\psi + U_\chi$, Fig.7, and of $V_\psi + V_\chi$, Fig.8, are found to be exactly identical to those of observed U and observed V respectively. Relative contributions of U_ψ and U_χ in U and V_ψ and V_χ in V are looked into from variance considerations.

Variances of any two distributions, say "E" and "F", viz. Var(E) and Var(F), are related to the variance of the total distribution, Var(E+F), (see APPENDIX-A) in the following relational form.

$$\mathrm{Var}(E+F) = \mathrm{Var}(E) + \mathrm{Var}(F) + 2\,\mathrm{Cov}(E,F) \qquad (14)$$

Where Cov(E,F) is the covariance between the distributions E and F. The algorithm presented here was applied to the total winds constituted from the monsoon seasonal (June - August) normal zonal and meridional winds at 1000, 850, 700, 500, 400, 200, 150 and 100 mb levels from the published charts of Newell et. al. (1972) in the domain (30 S,30 N),(60 E,145 E) having a 5 degree square grid. Vertical (pressure) profiles of the percentage of variance of U explained by (1) variance of U_ψ, (2) variance of U_χ, (3) twice the covariance between U_ψ and U_χ and (4) variance of $U_\psi + U_\chi$ are shown in Fig.9 while Fig.10 depicts the corresponding profiles for the meridional components. The following could be inferred from Figs.9 and 10.

(1) Zonal and meridional components viz. $U_\psi + U_\chi$ and $V_\psi + V_\chi$ of the recomputed wind explain 100% of the variance of the corresponding components of the observed wind at all levels.

(2) Percentage of variance of observed U explained by U_ψ (rota-

tional part) is predominantly more than that explained by

U_χ (divergent part) at all levels.

(3) Percentage of variance of observed V explained by (a) V_ψ (rotational part) is more than that explained by V_χ (divergent part) at levels above 950 mb and below 400 mb (b) V_χ (divergent part) is more than that explained by V_ψ (rotational part) in the pressure belt 400/950 mb.

(4) Percentage of variance of observed V explained by $Cov(V_\psi, V_\chi)$ is less than the variances explained by the component fields at all pressure levels while this is not so in the case of the zonal component U.

Interdependence of one flow on the other is studied by studying the variations of the correlation coefficients between the apopropriate fields at different isobaric levels. The correlation coefficient, Corr(E,F), between any two distributions "E" and "F" is a symbolic representative of the qualitative variations in one distribution influenced by the variations in the other distribution and is defined by the following relation.

$$Corr(E,F) = Cov(E,F)/Sqrt(Var(E)Var(F)) \qquad (15)$$

Values of the correlation coefficients, $Corr(U_\psi, U_\chi)$ and $Corr(V_\psi, V_\chi)$ at different isobaric levels for the data set scanned are tabulated in Table 1. The following could be inferred from the information of Table 1.

(1) U_ψ (rotational part) and U_χ (divergent part) are posit-

ively correlated at all levels with a more positive correlation at 400 mb level.

(2) V_ψ (rotational part) and V_χ (divergent part) are negatively correlated from surface to 300 mb level (except at 850 mb) with more negative correlation at 400/500 mb.

(3) V_ψ (rotational part) and V_χ (divergent part) are positively correlated at and above 200 mb.

The results presented in this algorithm are pertinent to the data set investigated. The magnitude of the variances, correlation coefficients and covariances may vary depending on the data set investigated.

Acknowledgements.

The author thanks the Director, I.I.T.M., for providing the facilities for carrying out this work. Thanks are due to Dr. S.K. Mishra, Deputy Director, for encouragement. Author also thanks Drs. S.Rajamani and (Mrs.) P.S.Salvekar for reviewing this paper and offering constructive suggestions. Author also thanks the "Drawing Section" of IITM for preparing tracings of this work. Utilisation of ND-560/Cx computer facility is thankfully acknowledged.

++++++++++++++++++++++++++

References

Bijlsma, S.J., Hafkenscheid, L.M., Lynch Peter : Computation of the streamfunction and velocity potential and reconstruction of the wind field. Mon.Wea.Rev., 114, 1547-1551, 1986.

Newell, R.E., Kidson, J.W., Vincent, D.G., Boer, G.J. : The General Circulation of the Tropical Atmosphere and Interactions with Extratropical Latitudes. Vol. I, The MIT Press, pp. 258 and pp. 370, 1972.

Sardeshmukh, P.D., and Hoskins, B.J., : On the derivation of the divergent flow from rotational flow : The Ki problem. Q.J.R. M.S., 113, 339-360, 1987.

Stephens, J.J., Johnson, K.W. : Rotational and Divergent Wind Potentials. Mon,Wea.Rev., 108, 1452-1457, 1978.

Tandon, M.K. : A FORTRAN Algorithm for 2-Dimensional Harmonic Analysis ,IITM Research Report No. R-047, Nov. 1991.

++++++++++++++++++++++++++

APPENDIX-A

The variance of any distribution B, viz. Var(B), is defined as

$$Var(B) = \langle (B - \langle B \rangle)^2 \rangle \qquad (A.1)$$

which on simplification becomes as

$$Var(B) = \langle B^2 \rangle - \langle B \rangle^2 \qquad (A.2)$$

where <()> denotes mean of (). Now if the given distribution B is expressed as the combination of two other distributions "E" and "F" such that B = E + F then the Eq.(A.2) becomes

$$Var(E+F) = \langle (E + F)^2 \rangle - \langle E + F \rangle^2 \qquad (A.3)$$

$$= \langle E^2 \rangle - \langle E \rangle^2 + \langle F^2 \rangle - \langle F \rangle^2$$

$$+ 2(\langle EF \rangle - \langle E \rangle \langle F \rangle) \qquad (A.4)$$

Eq.(A.4) with Eq.(A.2) gives us the following relation.

$$Var(E+F) = Var(E) + Var(F) + 2\ Cov(E,F) \qquad (A.5)$$

where

$$Cov(E,F) = \langle EF \rangle - \langle E \rangle \langle F \rangle \qquad (A.6)$$

is the covariance between the distributions "E" and "F". Equivalents of Eq.(A.5) in the context of presernt study are the following.

$$Var(U_\psi + U_\chi) = Var(U_\psi) + Var(U_\chi) + 2\ Cov(U_\psi, U_\chi) \qquad (A.7)$$

$$Var(V_\psi + V_\chi) = Var(V_\psi) + Var(V_\chi) + 2\ Cov(V_\psi, V_\chi) \qquad (A.8)$$

Eqs(A.7) and (A.8) are used in analysing the interdepence of rotational and divergent flows in the present study.

++++++++++++++++++++++++++

APPENDIX-B

```
C
C         Listing of (FORTRAN-77) Algorithm - routines
C
C -----------------------------------------------------------------
C
C     .....  Description of variables   .....
C
C            ... Scalar variables ...
C
C      PI     :  3.1415926536
C      NX     :  Number of grid points along x-direction.
C      NY     :  Number of grid points along y-direction.
C      NCX    :  Value of truncation along x-direction.
C      NCY    :  Value of truncation along y-direction.
C
C      NDM1   :  An intermediate variable.
C      NDM2   :  An intermediate variable.
C      NDM3   :  An intermediate variable.
C      NXHFP  :  An intermediate variable.
C      NYHFP  :  An intermediate variable.
C      DANGX  :  An intermediate variable.
C      DANGY  :  An intermediate variable.
C
C             ... Array variables ...
C
C      BCFSX         :  An array storing angles associated with
C                       cosine term of 1-D fourier series along
C                       x-direction.
C
C      BSFSX         :  An array storing angles associated with
C                       sine term of 1-D fourier series along
C                       x-direction.
C
C      BCFSY         :  An array storing angles associated with
C                       cosine term of 1-D fourier series along
C                       y-direction.
C
```

```
C          BSFSY            :  An array storing angles associated with
C                              sine term of 1-D fourier series along
C                              y-direction.
C
C          BCSFSX(M,N,1) :  An array storing summation of angles
C                              associated with cosine term of 1-D
C                              fourier series along x-direction.
C
C          BCSFSX(M,N,2) :  An array storing summation of angles
C                              associated with sine term of 1-D
C                              fourier series along x-direction.
C
C          BCSFSY(M,N,1) :  An array storing summation of angles
C                              associated with cosine term of 1-D
C                              fourier series along y-direction.
C
C          BCSFSY(M,N,2) :  An array storing summation of angles
C                              associated with sine term of 1-D
C                              fourier series along y-direction.
C
C          UCC              :
C          UCS              :  These are the expansion coefficients of the
C          USC              :  component of original total wind.
C          USS              :
C
C          VCC              :
C          VCS              :  These are the expansion coefficients of the
C          VSC              :  meridional component of original total wind.
C          VSS              :
C
C          UCYCC            :
C          UCYCS            :  These are the expansion coefficients of the
C          UCYSC            :  zonal component ,UCY, of rotational wind.
C          UCYSS            :
C
C          UKYCC            :
C          UKYCS            :  These are the expansion coefficients of the
C          UKYSC            :  zonal component ,UKY, of divergent wind.
C          UKYSS            :
C
```

```
C       VCYCC        :
C       VCYCS        :  These are the expansion coefficients of the
C       VCYSC        :  meridional component ,VCY, of rotational wind.
C       VCYSS        :
C
C       VKYCC        :
C       VKYCS        :  These are the expansion coefficients of the
C       VKYSC        :  meridional component ,VKY, of divergent wind.
C       VKYSS        :
C
C       UCY          :  Grid point distribution of UCY.
C       UKY          :  Grid point distribution of UKY.
C       VCY          :  Grid point distribution of VCY.
C       VKY          :  Grid point distribution of VKY.
C
C       UT           :  Grid point distribution of UCY + UKY ; recomputed
C                    :  zonal component of total wind.
C
C       VT           :  Grid point distribution of VCY + VKY ; recomputed
C                       meridional component of total wind.
C
C                ..... Description of routines  .....
C
C
C       CONST   :  It defines intermediate variables and computes
C                  array variables BCFSX , BSFSX , BCFSY , BSFSY.
C
C       ANGLES  :  It computes array variables associated with 1-D
C                  fourier series.
C
C       SUMANG  :  It computes array variables BCSFSX , BCSFSY.
C
C       CFSDBL  :  It computes four 2-D harmonics from 2-D grid point
C                  distribution.
C
C       FORIER  :  It computes 1-D harmonics from 1-D grid point
C                  distribution.
C
C       SUMDBL  :  It generates 2-D grid point distribution using
C                  four 2-D harmonics computed from routine CFSDBL.
```

```
C
C       OBSERV  :  It computes variance of a 2-D grid point
C                  distribution.
C
C       COVAR   :  It computes the covariance of two fields.
C
C       BASIK   :  It computes the expansion coefficients of the zonal
C                  and meridional components of the original total
C                  wind (U,V).
C
C       BREAK   :  It computes the grid point distribution of the
C                  divergent and rotational wind fields using the
C                  expansion coefficients from "BASIK".
C
C
C                 .....  Description of common blocks   .....
C
C
C      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
C     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
C      COMMON/T1/UCS(16,16),UCC(16,16),USC(16,16),USS(16,16)
C      COMMON/T2/VCS(16,16),VCC(16,16),VSC(16,16),VSS(16,16)
C      COMMON/T3/UCYCS(16,16),UCYCC(16,16),UCYSC(16,16),UCYSS(16,16)
C      COMMON/T4/VCYCS(16,16),VCYCC(16,16),VCYSC(16,16),VCYSS(16,16)
C      COMMON/T5/UKYCS(16,16),UKYCC(16,16),UKYSC(16,16),UKYSS(16,16)
C      COMMON/T6/VKYCS(16,16),VKYCC(16,16),VKYSC(16,16),VKYSS(16,16)
C      COMMON/T7/UCY(31,31),UKY(31,31),VCY(31,31),VKY(31,31)
C      COMMON/T8/UT(31,31),VT(31,31)
C      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
C     *BCSFSY(31,16,2)
C      COMMON/T10/PSY(31,31),CHI(31,31)
C
C     ----------------------------------------------------------------
C

      SUBROUTINE CONST(DX,DY)
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      RADIUS=6.371E6
```

```
      PI=3.1415926536
      DEG=PI/180.*RADIUS
      WLX=DX*FLOAT(NX-1)*DEG
      WNX=2.*PI/WLX
      SQWNX=WNX*WNX
      WLY=DY*FLOAT(NY-1)*DEG
      WNYHF=2.*PI/WLY
      SWNYHF=WNYHF*WNYHF
      WNYBX=WNYHF/WNX
      SWNYBX=WNYBX*WNYBX
      WLX=.001*WLX
      WLY=.001*WLY
      NCX=NX/2+1
      NCY=NY/2+1
      NYHFP=NY/2+1
      NXHFP=NX/2+1
      CALL ANGLES(NCX,PI,NX,BCFSX,BSFSX)
      CALL ANGLES(NCY,PI,NY,BCFSY,BSFSY)
      CALL SUMANG
      RETURN
      END
      SUBROUTINE COVAR(B,C,NX,NY,COV)
C
C     Computations of 2*Cov(B,C).
C
      DIMENSION B(NX,NX),C(NY,NY)
      SB=0.
      SC=0.
      SBC=0.
      DO 1 M=1,NX
      DO 1 N=1,NY
      SB=SB+B(N,M)
      SC=SC+C(N,M)
    1 SBC=SBC+B(N,M)*C(N,M)
      ANXY=1./FLOAT(NX*NY)
      COV=2.*ANXY*(SBC-SB*SC*ANXY)
      RETURN
      END
      SUBROUTINE ANGLES(NCX,PI,NX,BCFS,BSFS)
      DIMENSION BCFS(NCX),BSFS(NCX)
```

```
      T=2.*PI/FLOAT(NX)
      C=COS(T)
      S=SIN(T)
      C1=1.
      S1=0.
      DO 1 L=1,NCX
      BCFS(L)=C1
      BSFS(L)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    1 S1=B
      RETURN
      END
      SUBROUTINE SUMANG
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      T=2.*PI/FLOAT(NX)
      ANG=0.
      DO 1 M=1,NXHFP
      ANG=ANG+T
      IF(M.EQ.NXHFP)ANG=T*FLOAT(NX)
      C=COS(ANG)
      S=SIN(ANG)
      C1=1.
      S1=0.
      DO 1 N=1,NCX
      BCSFSX(M,N,1)=C1
      BCSFSX(M,N,2)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    1 S1=B
      T=2.*PI/FLOAT(NY)
      ANG=0.
      DO 2 M=1,NYHFP
      ANG=ANG+T
      IF(M.EQ.NYHFP) ANG=T*FLOAT(NY)
```

```
      C=COS(ANG)
      S=SIN(ANG)
      C1=1.
      S1=0.
      DO 2 N=1,NCY
      BCSFSY(M,N,1)=C1
      BCSFSY(M,N,2)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    2 S1=B
      RETURN
      END
      SUBROUTINE CFSDBL(F,CC,CS,SC,SS)
C
C     This routine splits a 2-D grid point distribution into a double
C        fouries series & computes 2-D fourier expansion coefficients
C        CC(N,M),CS(N,M),SC(N,M) AND SS(N,M) using following series :-
C
C               NCX  NCY
C               ---  ---
C               \    \     CC(N,M)COS(MKX)COS(NLY) + CS(N,M)COS(MKX)SIN(NLY)
C     F(X,Y)=
C               /    /   + SC(N,M)SIN(MKX)COS(NLY) + SS(N,M)SIN(MKX)SIN(NLY)
C               ---  ---
C               M=1  N=1
C
C
      DIMENSION F(31,31),CC(16,16),CS(16,16),SC(16,16),SS(16,16),FC(31),
     *FS(31),FCC(16),FCS(16),FSC(16),FSS(16),G(31),DC(31,16),DS(31,16)
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      DO 5 J=1,NX
      DO 1 K=1,NY
    1 G(K)=F(J,K)
      CALL FORIER(G,FCC,FSS,NCY,NY,BCFSY,BSFSY)
      DO 5 K=1,NCY
      DC(J,K)=FCC(K)
```

```
 5 DS(J,K)=FSS(K)
   DO 7 K=1,NCY
   DO 6 J=1,NX
   FC(J)=DC(J,K)
 6 FS(J)=DS(J,K)
   CALL FORIER(FC,FCC,FCS,NCX,NX,BCFSX,BSFSX)
   CALL FORIER(FS,FSC,FSS,NCX,NX,BCFSX,BSFSX)
   DO 7 J=1,NCX
   CC(J,K)=FCC(J)
   CS(J,K)=FCS(J)
   SC(J,K)=FSC(J)
 7 SS(J,K)=FSS(J)
   RETURN
   END
   SUBROUTINE FORIER(X,Y,Z,NC,NDATA,BCFS,BSFS)
C
C     First data point chosen as origin.
C
   DIMENSION X(NDATA),Y(NC),Z(NC),BCFS(NC),BSFS(NC)
   MAXFC=NDATA/2+1
   NDM1=NDATA-1
   NDM2=NDATA-2
   NDM3=NDATA-3
   AL=NDATA
   FFACR=2./AL
   CSUM=X(1)
   DO 1 I=2,NDATA
 1 CSUM=CSUM+X(I)
   CSUM=CSUM/AL
   DO 2 I=1,NDATA
 2 X(I)=X(I)-CSUM
   Y(1)=CSUM
   Z(1)=0.
   DO 8 M=2,NC
   C1=BCFS(M)
   S1=BSFS(M)
   UKP2=X(NDM1)
   UKP1=X(NDM2)+(C1+C1)*UKP2
   K=NDM2
   DO 7 I=1,NDM3
```

```
      K=K-1
      UK=X(K)+(C1+C1)*UKP1-UKP2
      UKP2=UKP1
    7 UKP1=UK
      Z(M)=FFACR*UK*S1
    8 Y(M)=FFACR*(X(NDATA)+UKP1*C1-UKP2)
      IF((MOD(NDATA,2).EQ.0).AND.(NC.EQ.MAXFC)) Y(NC)=.5*Y(NC)
      RETURN
      END
      SUBROUTINE SUMDBL(A,B,C,D,F)
C
C     Suming a truncated double fourier series.
C
      DIMENSION A(16,16),B(16,16),C(16,16),D(16,16),F(31,31),G(31,16),H(
     *31,16)
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      DO 2 N=1,NCY
      DO 2 MX=1,NXHFP
      S=0.
      T=0.
      U=0.
      V=0.
      DO 1 M=1,NCX
      U=U+D(N,M)*BCSFSX(MX,M,2)
      V=V+C(N,M)*BCSFSX(MX,M,1)
      S=S+B(N,M)*BCSFSX(MX,M,2)
    1 T=T+A(N,M)*BCSFSX(MX,M,1)
      IF(MX-NXHFP) 5,6,6
    5 G(NX-MX,N)=T-S
      H(NX-MX,N)=V-U
      G(MX,N)=T+S
      H(MX,N)=V+U
      GO TO 2
    6 G(NX,N)=T+S
      H(NX,N)=V+U
    2 CONTINUE
      DO 3 MX=1,NX
```

```
      DO 3 MY=1,NYHFP
      S=0.
      T=0.
      DO 4 M=1,NCY
      S=S+H(MX,M)*BCSFSY(MY,M,2)
    4 T=T+G(MX,M)*BCSFSY(MY,M,1)
      IF(MY-NYHFP) 7,8,8
    7 F(MX,NY-MY)=T-S
      F(MX,MY)=T+S
      GO TO 3
    8 F(MX,NY)=T+S
    3 CONTINUE
      RETURN
      END
      SUBROUTINE OBSERV(F,NX,NY,VAROB)
C
C     Variance computations from grid point data.
C
      DIMENSION F(NX,NY)
      S=0.
      DO 1 M=1,NX
      DO 1 N=1,NY
    1 S=S+F(N,M)
      S=S/(FLOAT(NX)*FLOAT(NY))
      VAROB=0.
      DO 2 M=1,NX
      DO 2 N=1,NY
    2 VAROB=VAROB+(F(N,M)-S)*(F(N,M)-S)
      VAROB=VAROB/(FLOAT(NX)*FLOAT(NY))
      RETURN
      END
      SUBROUTINE BASIK
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T0/U(31,31),V(31,31)
      COMMON/T1/UCS(16,16),UCC(16,16),USC(16,16),USS(16,16)
      COMMON/T2/VCS(16,16),VCC(16,16),VSC(16,16),VSS(16,16)
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      CALL CFSDBL(U,UCC,UCS,USC,USS)
```

```
      CALL CFSDBL(V,VCC,VCS,VSC,VSS)
      RETURN
      END
      SUBROUTINE BREAK
C
C     Generating
C               (1) Rotational (non-divergent)
C           and (2) Divergent  (ir-rotational)
C           wind fields.
C
      REAL KYCC,KYCS,KYSC,KYSS
      DIMENSION CHICC(16,16),CHICS(16,16),CHISC(16,16),CHISS(16,16),PSYC
     *C(16,16),PSYSC(16,16),PSYSS(16,16),PSYCS(16,16)
      COMMON PI,WNX,SQWNX,WNYHF,SWNYHF,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,
     *WNYBX,SWNYBX,WLX,WLY,NCX,NCY
      COMMON/T1/UCS(16,16),UCC(16,16),USC(16,16),USS(16,16)
      COMMON/T2/VCS(16,16),VCC(16,16),VSC(16,16),VSS(16,16)
      COMMON/T3/UCYCS(16,16),UCYCC(16,16),UCYSC(16,16),UCYSS(16,16)
      COMMON/T4/VCYCS(16,16),VCYCC(16,16),VCYSC(16,16),VCYSS(16,16)
      COMMON/T5/UKYCS(16,16),UKYCC(16,16),UKYSC(16,16),UKYSS(16,16)
      COMMON/T6/VKYCS(16,16),VKYCC(16,16),VKYSC(16,16),VKYSS(16,16)
      COMMON/T7/UCY(31,31),UKY(31,31),VCY(31,31),VKY(31,31)
      COMMON/T8/UT(31,31),VT(31,31)
      COMMON/T9/BCFSX(16),BSFSX(16),BCFSY(16),BSFSY(16),BCSFSX(31,16,2),
     *BCSFSY(31,16,2)
      COMMON/T10/PSY(31,31),CHI(31,31)
      DO 1 M=1,NCX
      WM=FLOAT(M)*WNX
      DO 1 N=1,NCY
      WN=FLOAT(N)*WNYHF
      W=WM*WM+WN*WN
      W=1./W
      KYCC=-W*(WM*UCS(N,M)+WN*VSC(N,M))
      KYCS=-W*(-WM*UCC(N,M)+WN*VSS(N,M))
      KYSC=-W*(WM*USS(N,M)-WN*VCC(N,M))
      KYSS=W*(WM*USC(N,M)+WN*VCS(N,M))
      CYCC=-W*(WM*VCS(N,M)-WN*USC(N,M))
      CYCS=W*(WM*VCC(N,M)+WN*USS(N,M))
      CYSC=-W*(WM*VSS(N,M)+WN*UCC(N,M))
      CYSS=-W*(WN*UCS(N,M)-WM*VSC(N,M))
```

```
      CHICC(N,M)=KYCC
      CHICS(N,M)=KYCS
      CHISC(N,M)=KYSC
      CHISS(N,M)=KYSS
      PSYCC(N,M)=CYCC
      PSYCS(N,M)=CYCS
      PSYSC(N,M)=CYSC
      PSYSS(N,M)=CYSS
      UCYCC(N,M)=-WN*CYSC
      UCYCS(N,M)=-WN*CYSS
      UCYSC(N,M)=WN*CYCC
      UCYSS(N,M)=WN*CYCS
      VCYCC(N,M)=WM*CYCS
      VCYCS(N,M)=-WM*CYCC
      VCYSC(N,M)=WM*CYSS
      VCYSS(N,M)=-WM*CYSC
      UKYCC(N,M)=WM*KYCS
      UKYCS(N,M)=-WM*KYCC
      UKYSC(N,M)=WM*KYSS
      UKYSS(N,M)=-WM*KYSC
      VKYCC(N,M)=WN*KYSC
      VKYCS(N,M)=WN*KYSS
      VKYSC(N,M)=-WN*KYCC
    1 VKYSS(N,M)=-WN*KYCS
      CALL SUMDBL(CHICC,CHICS,CHISC,CHISS,CHI)
      CALL SUMDBL(PSYCC,PSYCS,PSYSC,PSYSS,PSY)
      CALL SUMDBL(UCYCC,UCYCS,UCYSC,UCYSS,UCY)
      CALL SUMDBL(VCYCC,VCYCS,VCYSC,VCYSS,VCY)
      CALL SUMDBL(UKYCC,UKYCS,UKYSC,UKYSS,UKY)
      CALL SUMDBL(VKYCC,VKYCS,VKYSC,VKYSS,VKY)
      DO 2 M=1,NX
      DO 2 N=1,NY
      UT(N,M)=UCY(N,M)+UKY(N,M)
    2 VT(N,M)=VCY(N,M)+VKY(N,M)
      RETURN
      END
```

++++++++++++++++++++++++++++

Table 1

Numerical values of $Var(U_\psi)$, $Var(U_\chi)$, $Var(V_\psi)$, $Var(V_\chi)$, $Cov(U_\psi, U_\chi)$, $Cov(V_\psi, V_\chi)$, $Corr(U_\psi, U_\chi)$ and $Corr(V_\psi, V_\chi)$. All variances and covariances are the in units of meter square per second square.

mb level	$Var(U_\psi)$	$Var(U_\chi)$	Cov (U_ψ, U_χ)	Corr (U_ψ, U_χ)	$Var(V_\psi)$	$Var(V_\chi)$	Cov (V_ψ, V_χ)	Corr (V_ψ, V_χ)
1000	4.323	1.233	1.036	0.4487	0.8054	1.2628	-0.3434	-0.3401
850	17.183	1.580	1.770	0.3397	2.4570	1.8372	0.0230	0.0108
700	10.513	1.323	1.447	0.3880	2.3920	1.3920	-0.6130	-0.3359
500	38.672	1.307	4.148	0.5833	1.4150	1.1770	-0.7380	-0.5718
400	88.020	1.847	9.268	0.7262	1.5020	1.3450	-0.8110	-0.5705
300	181.575	4.845	19.935	0.6719	1.7880	2.5580	-0.7410	-0.3465
200	265.569	17.711	33.084	0.4823	2.2590	6.4790	0.9280	0.2425
150	258.711	19.330	32.871	0.4648	3.6890	6.1450	1.3160	0.2704
100	161.687	6.269	18.138	0.5694	2.0680	4.4080	0.2390	0.0784

++++++++++++++++++++++++++



Fig.1 : Isopleths of the zonal component U of the original (INPUT) total wind field $(U,V) = \hat{i}\, U + \hat{j}\, V$, in units of m/sec.



Fig.2 : Isopleths of the computed zonal component of the rotational wind field U_ψ , in units of m/sec.



Fig.3 : Isopleths of the computed zonal component of the divergent wind field; U_{χ} in units of m/sec.



Fig.4 : Isopleths of the meridional component V of the original (INPUT) total wind field $(U, V) = \hat{i}\, U + \hat{j}\, V$, in units of m/sec.



Fig.5 : Isopleths of the computed meridional component of the rotational wind field, V_{ψ} , in units of m/sec.



Fig.6 : Isopleths of the computed meridional component of the divergent wind field, V_χ , in units of m/sec.



Fig.7 : Isopleths of the zonal component of the recomputed total wind field, $u_\psi + u_\chi$, in units of m/sec.



Fig.8 : Isopleths of the meridional component of the recomputed total wind field $V_\psi + V_\chi$, in units of m/sec.



Fig.9 : Vertical profiles of the percentage variance of the variance of observed U explained by

(a) the variance of U_ψ ,

(b) the variance of U_χ ,

(c) 2 Cov (U_ψ , V_χ),

(d) sum of (a) and (b).



Fig.10 : Vertical profiles of the percentage variance of the variance of the variance observed V explained by (a) the variance of V_ψ (b) the variance of V_χ (c) 2 Cov (V_ψ , V_χ) (d) Sum of (a) and (b).